MODELOS DIDÁCTICOS PARA MECÁNICA RACIONAL EMPLEANDO MAPLE

José María Goicolea y Juan José Arribas,

E.T.S. Ingenieros de Caminos, Universidad Politécnica de Madrid, Ciudad Universitaria, 28040 Madrid

    e-mail: goico@caminos.upm.es
www: http://filemon.caminos.upm.es

Resumen

Los programas de matemática simbólica como Maple pueden constituir una ayuda valiosa para la docencia de la mecánica, facilitando el dominio del aparato matemático preciso y consiguiendo además resultados analíticos y numéricos que serían inabordables manualmente.

Se describe aquí el trabajo realizado en nuestra cátedra con alumnos de 2^o curso de ingeniería. Se expone asimismo de forma detallada un ejemplo de aplicación que incluye varias técnicas fundamentales de la dinámica de sistemas discretos (ecuaciones de Lagrange, ecuaciones de Newton-Euler, resolución numérica), incluyendo los resultados y las instrucciones de Maple necesarias para cada una de ellas.

1. INTRODUCCIÓN

La mecánica racional en las carreras de ingeniería constituye una materia intermedia entre las matemáticas y otras materias aplicadas (mecánica de medios continuos, cálculo de estructuras, mecánica de suelos y rocas, mecánica de fluidos, cinemática y dinámica de máquinas, . . . ). Como tal cumple un doble papel: por una parte afianzar sólidamente conceptos básicos para aplicaciones posteriores, y por otra aplicar los modelos matemáticos para la resolución de problemas. Este último aspecto constituye una componente esencial en la formación del ingeniero generalista.

Es precisamente el empleo de métodos matemáticos de cierta complejidad lo que constituye simultáneamente una ventaja y un inconveniente en esta materia. Ventaja puesto que se busca entrenar al estudiante en la resolución de problemas mediante modelos matemáticos, con base en conceptos físicos. Inconveniente puesto que en ocasiones la complejidad de los modelos y el aparato matemático dificultan la comprensión intuitiva de la mecánica por parte del alumno.

Los programas de matemáticas por ordenador existentes en la actualidad constituyen una potente ayuda que puede ser utilizada para solventar el inconveniente antes mencionado. Estos programas permiten, sin necesidad de emplear lenguajes de programación, una aplicación rápida de los modelos de cálculo de forma interactiva y “amigable” con el usuario. Debido a esta característica facilitan la resolución de gran número de problemas con un tiempo de aprendizaje pequeño. Este no es el caso de los lenguajes de programación (Fortran, C, C++, o incluso

Pascal o Basic) cuya finalidad principal suele ser la de obtener un programa eficaz para su uso reiterado con posterioridad, en detrimento de la interactividad.

Entre los programas de matemáticas por ordenador cabe diferenciar dos grandes grupos:

1. Los programas de cálculo numérico, como Mathcad, Matlab u otros. Estos sirven para evaluar numéricamente expresiones y resolver ecuaciones, realizar gráficos, etc. Entre estos cabría incluir también las hojas de cálculo como Lotus, Excel, etc.

2. Los programas de cálculo simbólico, como Derive, Mathematica, Macsyma, Maple [1]. Además de cálculos numéricos estos programas pueden operar de forma “simbólica” en gran cantidad de modelos matemáticos.

El desarrollo analítico a nivel “simbólico”, sin necesidad de evaluar las expresiones mediante valores numéricos es en ocasiones esencial para comprender la génesis de ciertos resultados y sus características. La evolución constante del “software” ha conducido a que algunas versiones modernas de programas “numéricos” incluyan también ciertas prestaciones “simbólicas”, aunque por lo general no tan completas como los programas del segundo tipo.

Para completar esta descripción debe decirse también que ambos tipos de programas suelen incorporar lenguajes sencillos de programación (interpretada, no compilada) así como capacidades de representación gráfica, animación, etc.

En lo que resta de este artículo describimos en primer lugar los objetivos y experiencia docente de nuestra cátedra empleando el programa de matemática simbólica Maple. A continuación se desarrolla en detalle una aplicación típica en la que intervienen varios conceptos y métodos fundamentales de la materia, incluyéndose las instrucciones necesarias de Maple V3 y los resultados obtenidos. Por último se resumen las conclusiones principales que extraemos de nuestra experiencia y se comenta la previsible evolución de estos métodos en las enseñanzas técnicas.

2. EL ORDENADOR EN LA MECÁNICA DE 2^o CURSO

La mecánica de segundo curso en una escuela de ingeniería suele precisar la aplicación de modelos matemáticos de cierta complejidad para problemas de cinemática, dinámica o estática. Estos requieren el empleo de vectores y tensores, álgebra matricial, cálculo diferencial e integral, ecuaciones diferenciales ordinarias, resolución de ecuaciones algebraicas lineales y no lineales, etc.

En prácticamente todos los bloques del temario de mecánica se pueden aplicar programas de matemáticas por ordenador como ayuda para el desarrollo de los conceptos y para la resolución de problemas de prácticas. A pesar de esto, la docencia de la asignatura y los métodos de evaluación continuan estando basados en técnicas “tradicionales”: la pizarra para las explicaciones teóricas del profesor, y el lápiz y papel para los ejercicios y exámenes del alumno.

En nuestra cátedra, el empleo del ordenador se hace mediante un seminariotaller voluntario específico al que atienden 125 alumnos cada curso. En él se emplea el programa Maple V3 para la resolución de problemas de mecánica. Se selecciona un subconjunto de aplicaciones suficientemente variado y significativo para proporcionar al alumno un entrenamiento adecuado que a la vez le facilite la comprensión de ciertos conceptos esenciales de la asignatura.

Los ejercicios que se resuelven en el seminario se seleccionan preferentemente entre problemas de examen, lo que contribuye a motivar al alumno al brindarle un punto de vista distinto para un problema que le resulta “familiar”. Se hace hincapié en obtener resultados adicionales sobre lo que se pediría en el examen para ser calculado a mano. Por ejemplo, es posible la resolución de sistemas de ecuaciones diferenciales ordinarias no lineales, lo que permite que el alumno comprenda la potencia del ordenador en aplicaciones para las que conoce bien las limitaciones del cálculo manual.

La motivación del estudiante constituye uno de los objetivos fundamentales de nuestra actividad, aspecto en el que el ordenador puede ayudar decisivamente. A menudo la ingeniería se ha basado en modelos matemáticos de corte excesivamente formal que el alumno considera como “obstáculos” a salvar, en lugar de “herramientas” que le faciliten la aplicación de una técnica determinada.

Por otra parte, la enseñanza debe adecuarse a la realidad de la práctica de la ingeniería, en la que el ordenador como herramienta de cálculo es omnipresente. Por este motivo y los anteriores consideramos que constituiría en gran medida un fraude al alumno no brindarle una enseñanza específica de los modelos de cálculo por ordenador más avanzados.

Sin embargo, como se ha dicho antes, la organización actual de la asignatura en nuestra cátedra no se basa exclusivamente en el ordenador sino que éste se incluye como una actividad adicional. Esto se debe a dos motivos principales:

1. Un cierto conservadurismo por el cual dudamos aún que sea conveniente basar la enseñanza exclusivamente en el ordenador, al menos mientras no exista una experiencia adecuada de sus resultados prácticos (medidos en cuanto a la capacitación profesional de los ingenieros y la versatilidad de su formación).

2. La escasez de recursos existentes que impedirían ofrecer a todos los alumnos la docencia adecuada y acceso a ordenadores y programas (La matrícula actual en nuestra asignatura es de más de 1000 alumnos).

Independientemente de estas consideraciones de tipo práctico, debidas a la imprescindible prudencia como educadores, estamos convencidos de que los métodos docentes evolucionarán en el sentido de involucrar crecientemente los ordenadores en la enseñanza. Es por tanto ineludible la introducción de lo que se ha denominado genéricamente “Computer Aided Teaching” (CAT). Asimismo es necesaria un experimentación amplia y un desarrollo de experiencias y materiales didácticos.

Nuestro trabajo se orienta en esta última línea, mediante un texto [3] que incluye materiales didácticos para las distintas partes del programa de la asignatura [2]:

1. dinámica de la partícula;

2. cinemática de campos de velocidades y aceleraciones; 3. dinámica de sistemas;

4. dinámica analítica de Lagrange;

5. cinemática y cinética del sólido rígido (rotaciones finitas, tensor de inercia);

6. dinámica del sólido;

7. impulsiones;

8. oscilaciones con n grados de libertad; 9. dinámica analítica de Hamilton; 10. estática;

11. sistemas de cables.

Este texto incluye en cada tema un resumen de los conceptos y modelos teóricos, seguido de los modelos de Maple con instrucciones precisas para su resolución. Estas instrucciones están también incluidas en un diskette.

3. APLICACIÓN A LA DINÁMICA DE SISTEMAS (ECUACIONES DE LAGRANGE Y DE NEWTON-EULER)

Con objeto de concretar la discusión genérica realizada arriba, desarrollamos aquí diversos modelos de cálculo mediante Maple que permiten ver las ventajas y limitaciones que poseen las distintas formulaciones para el estudio de la dinámica de sistemas.

Problema: Un semiaro de masa M y radio R rueda sin deslizar sobre una recta horizontal, manteniendose vertical en todo instante. Sobre él se mueve sin rozamiento una partícula de masa M con ligadura bilateral que no estorba la rodadura. Se emplearán como parámetros los ángulos θ y ϕ de giro del semiaro y de la partícula relativa al semiaro, ambos medidos desde la posición de equilibrio y en sentido antihorario. Se pide:

θ G

P ϕ

1. Lagrangiana del sistema. 2. Ecuaciones de Lagrange. 3. Integrales primeras.

4. Reacciones del semiaro sobre la recta (R_x, R_y) y la partícula (R_n) en el instante inicial para las condiciones de partida siguientes: (a) (θ_0 = 0, ϕ_0 = π/2, θ̇_0 = ϕ̇_0 = 0); (b) (θ_0 = π/2, ϕ_0 = π/2, θ̇_0 = ϕ̇_0 = 0); (c) (θ_0 = π/2, ϕ_0 = 0 θ̇_0 = ϕ̇_0 = 0).

5. Integrar numéricamente las ecuaciones obteniendo las ecuaciones horarias (θ(t), ϕ(t)). 6. Historias temporales de las fuerzas de reacción (R_x(t), R_y(t), R_n(t)).

1.- Se escoge en primer lugar la formulación analítica de Lagrange porque resulta enormemente ventajosa para formular las ecuaciones del movimiento directamente en función de las coordenadas libres (θ, ϕ). Se parte para ello de la expresión de la Lagrangiana: L = 1 2 M(ẋ^2_G + ẏ^2_G) + 1 2 J_Gθ̇^2 + 1 2 M(ẋ^2_P + ẏ^2_P ) − Mg(y_G + y_P )

donde (x_G, y_G) son las coordenadas del C.D.M. del semiaro, (x_P , y_P ) de la partícula, y J_G el momento de inercia del semiaro. Las instrucciones de Maple se incluyen en la tabla 1.

> restart:
> xG := -R*theta(t)+(2*R/Pi)*sin(theta(t));
> yG := R-(2*R/Pi)*cos(theta(t));
> xGt := diff(xG,t);
> yGt := diff(yG,t);
> xP := -R*theta(t)+R*sin(theta(t)+phi(t));
> yP := R-R*cos(theta(t)+phi(t));
> xPt := diff(xP,t);
> yPt := diff(yP,t);
> JG := M*R^2-M*(2*R/Pi)^2:
> T := (1/2)*M*(xGt^2+yGt^2)+(1/2)*JG*diff(theta(t),t)^2
      +(1/2)*M*(xPt^2+yPt^2);
> V := M*g*(yG+yP);
> L := T-V;

Tabla 1: Instrucciones de Maple para obtener la Lagrangiana

2.- Las ecuaciones de Lagrange se obtienen directamente mediante derivación de la Lagrangiana: d dt ( ∂L ∂q̇_i ) − ∂L ∂q_i = 0 (q_1 = θ, q_2 = ϕ)

Empleando el código que incluye en la tabla 2 resulta:

0 = 4MR^2θ̈ ( 1 − 1 π cos θ ) + MR^2θ̇^2 2 π sen θ

+MR^2[ϕ̈ − (2θ̈ + ϕ̈ cos(θ + ϕ) + (θ̇ + ϕ̇)^2 sen(θ + ϕ)] + MgR [ 2 π sen θ + sen(θ + ϕ) ]

0 = MR^2[(θ̈ + ϕ̈) − θ̈ cos(θ + ϕ)] + MgR sen(θ + ϕ)

3.- Sólo cabe citar la integral primera de la conservación de la energía (T + V = cte), expresión de obtención inmediata en Maple (> T+V;).

> Ltmp := subs({theta(t)=theta1,phi(t)=phi1},subs({diff(theta(t),t)=theta2,
  diff(phi(t),t)=phi2},L));
> tmp_0 := simplify(diff(Ltmp,theta2));
> tmp_1 := diff(subs({theta1=theta(t),phi1=phi(t)},
   subs({theta2=diff(theta(t),t),phi2=diff(phi(t),t)},tmp_0)),t);
> tmp_2 := subs({theta1=theta(t),phi1=phi(t)},
   subs({theta2=diff(theta(t),t),phi2=diff(phi(t),t)},
   simplify(diff(Ltmp,theta1))));
> ecl1_0 := tmp_1-tmp_2=0:
> ecl1 := collect(ecl1_0,
  [diff(theta(t),t,t),sin(theta(t)+phi(t)),cos(theta(t)+phi(t)),R,M],
  factor);
> tmp_0 := simplify(diff(Ltmp,phi2)):
> tmp_1 := diff(subs({theta1=theta(t),phi1=phi(t)},
   subs({theta2=diff(theta(t),t),phi2=diff(phi(t),t)},tmp_0)),t):
> tmp_2 := subs({theta1=theta(t),phi1=phi(t)},
   subs({theta2=diff(theta(t),t),phi2=diff(phi(t),t)},
   simplify(diff(Ltmp,phi1)))):
> ecl2_0 := tmp_1-tmp_2=0:
> ecl2 := collect(ecl2_0,
  [diff(theta(t),t,t),sin(theta(t)+phi(t)),cos(theta(t)+phi(t)),R,M],
  factor);

Tabla 2: Obtención de las ecuaciones de Lagrange. Obsérvense los cambios de variable (subs) auxiliares para realizar las derivadas parciales y las instrucciones empleadas para agrupar factores y simplificar expresiones (simplify, collect).

4.- Para obtener las reacciones en un instante dado lo más conveniente es aplicar las ecuaciones de Newton-Euler clásicas. En este caso se establecen 3 ecuaciones,

Mẍ_G = R_x + R_n sen(θ + ϕ) (1) Mÿ_G = R_y − Mg − R_n cos(θ + ϕ) (2) M[−Rθ̈ sen(θ + ϕ) − R(θ̇ + ϕ̇)^2] = −R_n + Mg cos(θ + ϕ) (3)

En la tabla 3 se detalla el código para despejar los valores de R_x, R_y y R_n de las ecuaciones (1), (2) y (3), eliminando también de ellas las aceleraciones mediante las ecuaciones de Lagrange obtenidas antes.

5.- La integración numérica de las ecuaciones de Lagrange se realiza de forma inmediata (tabla 4). Maple emplea por defecto un integrador Runge-Kutta de 4^o y 5^o orden. Es conveniente antes de manipular los resultados numéricos almacenarlos en tablas de valores.

En la figura 1 se muestra el dibujo realizado por Maple para la evolución de θ(t) y ϕ(t). Se observa en la figura que para aproximadamente t = 2.6 la partícula abandona el semiaro por el extremo opuesto al de entrada. Es posible también

> nu := array([sin(theta(t)+phi(t)),-cos(theta(t)+phi(t))]);
> ecN1 := M*diff(xGt,t)=Rx+Rn*nu[1];
> ecN2 := M*diff(yGt,t)=Ry-M*g+Rn*nu[2];
> ecN3 := simplify(M*(diff(xPt,t)*nu[1]+diff(yPt,t)*nu[2])=-Rn-M*g*nu[2]);
> Rns_0 := solve({ecN3},Rn):
> Rns := collect(Rns_0,
 [diff(theta(t),t,t),sin(theta(t)+phi(t)),
 cos(theta(t)+phi(t)),R,M],factor);
> Rxys_0 := simplify(solve(subs(Rns,{ecN1,ecN2}),{Rx,Ry})):
> Rxys := collect(Rxys_0,
  [diff(theta(t),t,t),sin(theta(t)+phi(t)),
  cos(theta(t)+phi(t)),R,M],factor);
> dds := simplify(solve({ecl1,ecl2},{diff(theta(t),t,t),diff(phi(t),t,t)}));
> Rxyss := subs(dds,Rxys);
> normal(eval(subs({theta(t)=Pi/2,phi(t)=Pi/2},
subs({diff(theta(t),t)=0,diff(phi(t),t)=0},
subs({diff(theta(t),t,t)=ddtheta_0,diff(phi(t),t,t)=ddphi_0},dds)))));
> normal(eval(subs({theta(t)=Pi/2,phi(t)=Pi/2},
subs({diff(theta(t),t)=0,diff(phi(t),t)=0},{Rns,Rxyss}))));

Tabla 3: Cálculo de aceleraciones y Reacciones en t = 0, para las condiciones iniciales θ_0 = ϕ_0 = π/2 (para otras condiciones bastaría con repetir las dos últimas instrucciones cambiando estos valores).

obtener estos gráficos en colores en pantalla, así como realizar animaciones gráficas empleando los datos del cálculo.

6.- Por último, manipulando las expresiones de las reacciones obtenidas antes se pueden particularizar con los resultados de la integración numérica y evaluar R_x(t), R_y(t) y R_n(t) (figura 2). En estos gráficos se aprecia cómo, una vez planteado y resuelto el problema de forma analítica por Maple, es posible obtener resultados numéricos que serían absolutamente inabordables de otra manera.

4. OTROS EJEMPLOS

Con objeto de ilustrar una aplicación distinta, se incluyen en la figura 3 las trayectorias dinámicas de una peonza simétrica. Se trata de un problema típico y de cierta complejidad de la dinámica de sólidos rígidos en tres dimensiones, para el cual se obtiene una visualización clara del comportamiento dinámico, lo que permite una comprensión intuitiva del mismo.

> M := 1:g := 9.8:R := 1:
> soln := dsolve({ecl1,ecl2,theta(0)=0,phi(0)=Pi/2,D(theta)(0)=0,D(phi)(0)=0},
  {theta(t),phi(t)},numeric);
> tmax := 3.0:npts := 200:dt := tmax/npts:
> soln_f := seq(soln(i*dt),i=0..npts):
> the_f := seq(subs(soln_f[i],theta(t)),i=1..npts+1):
> ph_f := seq(subs(soln_f[i],phi(t)),i=1..npts+1):
> dthe_f := seq(subs(soln_f[i],diff(theta(t),t)),i=1..npts+1):
> dph_f := seq(subs(soln_f[i],diff(phi(t),t)),i=1..npts+1):
> interface(plotdevice=postscript,plotoutput=‘figura1.eps‘,
  plotoptions=‘noborder‘);
> plot({[seq([(i-1)*dt,ph_f[i]],i=1..npts+1)],[seq([(i-1)*dt,
  the_f[i]],i=1..npts+1)]},t=0..tmax,phi=-Pi/2..Pi/2,title=‘theta(t) y phi(t)‘);

Tabla 4: Integración numérica de las ecuaciones y almacenamiento de tablas numéricas de valores. Se incluye también la instrucción que produce la figura 1.

Figura 1: Evolución de θ(t) y ϕ(t) resultado de la integración numérica. Obsérvese que el valor de |ϕ| llega a superar π/2, lo que significa que la partícula abandonaría el semiaro en ese instante.

Figura 2: Evolución de las reacciones R_x(t) y R_n(t) resultantes de la integración numérica.

a) b) c)

Figura 3: Los tres tipos de soluciones para el movimiento de la peonza simétrica, obtenidos mediante integración numérica en el ordenador de las ecuaciones del movimiento. La figura representa, sobre una superficie esférica, la trayectoria del extremo del eje de revolución de la peonza. En el caso a) la velocidad de precesión (ψ̇) lleva siempre el mismo signo; en el caso b) alterna de signo, describiendo bucles característicos; en el caso límite c), en los puntos de nutación mínima (puntos más altos de la trayectoria) se anula ψ̇, produciendo cúspides en la trayectoria.

5. CONCLUSIONES

La experiencia que se ha resumido arriba en el empleo docente de programas de matemáticas por ordenador nos conduce al siguiente resumen de conclusiones:

• Es imprescindible la introducción de programas de ordenador para auxiliar y facilitar la docencia de la mecánica teórica y aplicada, aunque esta introducción debe hacerse con la debida prudencia ya que no existe aún suficiente experiencia ni recursos didácticos.

• Los programas de matemática simbólica pueden ser una valiosa ayuda para facilitar al alumno el uso de las matemáticas como herramientas en lugar de convertirse en un obstáculos.

Referencias

[1] CHAR, B. N., y otros: Maple V, release 3, Languaje Reference Manual. Springer Verlag, 1991.

[2] GOICOLEA, J. M.: Mecánica. Servicio de publicaciones del Colegio de Inngenieros de Caminos (Colección Escuelas), 1995.

[3] GOICOLEA, J. M., ARRIBAS, J.J., y otros: Resolución de problemas de mecánica por ordenador mediante Maple, 1996.